

Lesson 1 Maple Basics

Maple is an interactive mathematical software package which combines, Numerical Computation, Symbolic Computation and Graphics.

Initializations

```
> restart;  
with(plots):
```

1.1 Numerical Computation

All arithmetic is exact, unless we explicitly ask for a decimal approximation using the **evalf** command.

Examples

Example 1.1.1

Rational Numbers, Large Integers, Trigonometric Values and Decimal Approximations.

```
> n1:=(2/7)^4;
```

$$n1 := \frac{16}{2401} \quad (2.1.1.1)$$

```
> n2:=evalf(n1);
```

$$n2 := 0.006663890046 \quad (2.1.1.2)$$

```
> n3:=73491097^6;
```

$$n3 := 157546362627224319599041649399034062211350126929 \quad (2.1.1.3)$$

```
> n4:=cos(Pi/6);
```

$$n4 := \frac{1}{2} \sqrt{3} \quad (2.1.1.4)$$

```
> n5:=evalf(n4);
```

$$n5 := 0.8660254040 \quad (2.1.1.5)$$

```
>
```

Example 1.1.2

Decimal Approximations of Arbitrary Length.

Decimal approximations of arbitrary length can be obtained by specifying the desired number of significant digits as an option in the **evalf** command. The default is ten significant digits.

```
> n6:=evalf(n4, 30);
```

$$n6 := 0.866025403784438646763723170755 \quad (2.1.2.1)$$

```
>
```

Example 1.1.3

Non-Trivial Trigonometric Values.

Some non-trivial trigonometric values can be found by using the **convert** to **radical** routine.

```
> n7:=cos(Pi/24);
```

$$n7 := \cos\left(\frac{1}{24} \pi\right) \quad (2.1.3.1)$$

```
> n8:=convert(n7, radical);
```

$$n8 := \frac{1}{4} \sqrt{8 + 2\sqrt{6} + 2\sqrt{2}} \quad (2.1.3.2)$$

```
>
```

Exercises

Exercise 1.1.1

Compute the exact value as well as a decimal approximation of $\sin\left(\frac{\pi}{12}\right)$.

```
>
```

1.2 Graphics


Maple contains an extensive array of plotting routines, most of which can be found in the **plots** package which is loaded using the **with(plots)** command which is conveniently placed in the Initializations group at the top of this worksheet.

Examples

Example 1.2.1

The plot command.

```
> plot(sin(x)/x, x=-10..10);
```




```
>
```

Example 1.2.2

Using the color option with the plot command.

We visualize the curve $x^3 - 2x^2$ and color it blue.

```
> plot(x^3-2*x^2, x=-2..3, color=blue);
```




```
>
```

Example 1.2.3

Limiting the range of a plot.

The rational expression $\frac{3x^2 + 5x - 4}{x^2 - 9}$ is not defined at $x = -3$ and $x = 3$. Moreover when x is close to either one of these values the absolute value of the expression becomes extremely large. In order to circumvent this complication, we limit the range of the plot with the option $y = -10..10$.

```
> plot((3*x^2+5*x-4)/(x^2-9), x=-7..7, y=-10..10);
```




```
>
```

Example 1.2.4

Multiple plots in one picture.

We can create multiple plots within one frame by using Maple set notation $\{ \dots \}$ or list notation $[\dots]$.


```
> plot({x^2, 3-2*x}, x=-4..3);
```



```
>
```

Set notation will work with any plot routine within Maple, list notation is only implemented for some. The advantage of list notation is that it allows for control of the color of individual curves.

```
> plot([x^2, 3-2*x], x=-4..3, color=[blue, red]);
```



```
>
```

▼ Example 1.2.5

Piecewise defined functions.

Code and display the graph of the function

$$f(x) = \begin{cases} 1 & x < 1 \\ 3-x & x \geq 1 \end{cases}$$

We use the **piecewise** command.


```
> f:=piecewise(x<1, 1, x>=1, 3-x);
```

$$f := \begin{cases} 1 & x < 1 \\ 3 - x & 1 \leq x \end{cases}$$

(3.1.5.1)

Observe that this function jumps at the point $x = 1$. We say that it is discontinuous at 1. Maple can be informed about this discontinuity by adding the **discont = true** option to the plot command.

```
> plot(f, x=-4..4, discont=true);
```




```
>
```

Example 1.2.6

Three-Dimensional images.

Maple can create three-dimensional graphs. Here is the picture of a saddlepoint surface.

```
> plot3d(x^2-y^2, x=-5..5, y=-5..5, style=patch, axes=boxed);
```



Exercises

Exercise 1.2.1

Sketch the graph of $f(x) = x^4 - x^2$ and use magenta as the color of the curve.

>

Exercise 1.2.2

Sketch the graph of $x \sin\left(\frac{1}{x}\right)$. Make sure that you choose the domain in such a way that the characteristics of the function around $x = 0$ are clearly displayed.

>

1.3 Symbolics

The real power of Maple lies in its symbolic capabilities. A computer algebra system allows the user to execute mathematical computations on the computer screen much like such they used to be performed with pencil and paper. This section explores some of the capabilities of the program.

Examples

Example 1.3.1

The Expand, Factor, and Sort commands.

The examples speak for themselves.

```

> e1:=(2*x-3*x^2)^2*(4*x-5)^3;
      e1 := (2 x - 3 x^2)^2 (4 x - 5)^3           (4.1.1.1)
> e2:=expand(e1);
      e2 := 5836 x^5 - 5685 x^4 + 2700 x^3 - 500 x^2 - 2928 x^6 + 576 x^7   (4.1.1.2)
> e3:=sort(e2, x);
      e3 := 576 x^7 - 2928 x^6 + 5836 x^5 - 5685 x^4 + 2700 x^3 - 500 x^2   (4.1.1.3)
> e4:=factor(e3);
      e4 := x^2 (3 x - 2)^2 (4 x - 5)^3         (4.1.1.4)
>

```

Example 1.3.2

Expressions and Functions.

All computer algebra systems make a clear distinction between expressions and functions. A function has to be associated with an argument (variable).

Maple syntax for an expression g is given by

$$g := \text{expression body};$$

Maple syntax for a function f of variable x is

$$f := x \rightarrow \text{function body};$$

A few examples say more than a thousand words.

We define the expression g equal to $x^2 + 7$ as well as the function $f(x) = x^2 + 7$. Then we evaluate f for $x = a$, $x = 2$ and $x = 5t + 1$.

```

> g:=x^2+7;
      g := x^2 + 7                               (4.1.2.1)
> f:=x->x^2+7;
      f := x -> x^2 + 7                          (4.1.2.2)
> f(a);
      a^2 + 7                                    (4.1.2.3)
> f(2);
      11                                         (4.1.2.4)
> f(5*t+1);
      (5 t + 1)^2 + 7                            (4.1.2.5)
>

```

Example 1.3.3

The solve command.

Solve the equation $ax^2 + bx + c = 0$.

```

> eq1:=a*x^2+b*x+c=0;
      eq1 := a x^2 + b x + c = 0                 (4.1.3.1)
> sol1:=solve(eq1, x);

```

(4.1.3.2)

$$\text{sol1} := -\frac{1}{2} \frac{b - \sqrt{b^2 - 4ac}}{a}, -\frac{1}{2} \frac{b + \sqrt{b^2 - 4ac}}{a} \quad (4.1.3.2)$$

>

Example 1.3.4

Systems of Equations.

Simultaneously solve the equations $3x - 4y = 1$ and $5x + 7y = -3$.

$$\begin{aligned} > \text{eq2} := \{3*x-4*y=1, 5*x+7*y=-3\}; \\ & \text{eq2} := \{3x - 4y = 1, 5x + 7y = -3\} \end{aligned} \quad (4.1.4.1)$$

$$\begin{aligned} > \text{sol2} := \text{solve}(\text{eq2}, \{x, y\}); \\ & \text{sol2} := \left\{y = -\frac{14}{41}, x = -\frac{5}{41}\right\} \end{aligned} \quad (4.1.4.2)$$

Exercises

Exercise 1.3.1

Compute the exact value and a decimal approximation of $\left(\frac{1231}{12}\right)^{11}$.

>

Exercise 1.3.2

Plot the graph of $\frac{x^2 - 2x + 5}{x^2 + 1}$.

>

Exercise 1.3.3

Plot in one picture the graphs of $f(x) = x \sin 2x$ and $g(x) = 5 - 3x^3$.

>

Exercise 1.3.4

Define the Maple function $f(x) = x^3 - 5x$ and compute $f(3u + 7)$.

>

Exercise 1.3.5

Solve the system of equations

$$\begin{aligned} 5x - 3y &= 11 \\ -2x + 8y &= 13 \end{aligned}$$

>